

4.2 Poisson Exact Test and Confidence Interval for Incidence Rate

Suppose that n cases have been observed during T person-years. We can assume that the number of cases in the population follow a Poisson distribution with mean λT , where λ is the true incidence rate. We might be interested in testing $H_0 : \lambda = \lambda_0$ against an upper-tailed $H_1 : \lambda > \lambda_0$, or lower-tailed $H_1 : \lambda < \lambda_0$, or two-tailed $H_1 : \lambda \neq \lambda_0$.

To compute the p -value for this test, we denote by N a Poisson random variable with the mean $\lambda_0 T$. The formula for the p -value depends on the sign in the alternative hypothesis:

- For $H_1 : \lambda > \lambda_0$, $p\text{-value} = \mathbb{P}(N \geq n) = \sum_{k=n}^{\infty} \frac{(\lambda_0 T)^k}{k!} e^{-\lambda_0 T}$.
- For $H_1 : \lambda < \lambda_0$, $p\text{-value} = \mathbb{P}(N \leq n) = \sum_{k=0}^n \frac{(\lambda_0 T)^k}{k!} e^{-\lambda_0 T}$.
- For $H_1 : \lambda \neq \lambda_0$, if $n < \lambda_0 T$,

$$p\text{-value} = \min \left\{ 2 \sum_{k=0}^n \frac{(\lambda_0 T)^k}{k!} e^{-\lambda_0 T}, 1 \right\}.$$

- For $H_1 : \lambda \neq \lambda_0$, if $n \geq \lambda_0 T$,

$$p\text{-value} = \min \left\{ 2 \sum_{k=n}^{\infty} \frac{(\lambda_0 T)^k}{k!} e^{-\lambda_0 T}, 1 \right\}.$$

Also, we can construct a $100 \cdot (1 - \alpha)\%$ **exact Poisson confidence interval** for the true incidence rate, $[\lambda_L, \lambda_U]$, where the confidence limits satisfy:

$$\sum_{k=n}^{\infty} \frac{(\lambda_L T)^k}{k!} e^{-\lambda_L T} = \alpha/2, \text{ and } \sum_{k=0}^n \frac{(\lambda_U T)^k}{k!} e^{-\lambda_U T} = \alpha/2.$$

Example. As introduced earlier, suppose 12 individuals got sick over the course of 1978 person-days of the study. The incidence rate in this case is $12/1,978 = 0.006067$ per person-day or roughly 6 per 1,000 person-days. Below we conduct an exact Poisson test to see if this incidence rate is less than 10 per 1,000 person-days, that is, we test $H_0 : \lambda = 0.01$ against the alternative $H_1 : \lambda < 0.01$,

In SAS:

```

/*Exact Poisson Test*/
data cases;
input T n;
cards;
1978 12
;

%let lambda0=0.01;
%let H1="less"; *choices: "less", "greater", "two.sided";

data test;
set cases;
if (&H1="greater") then
  pvalue=cdf("gamma", &lambda0*T, n);
  if (&H1="less") then
    pvalue=1-cdf("gamma", &lambda0*T, n+1);
if (&H1="two.sided" and n<&lambda0*T) then
  pvalue=min(2*(1-cdf("gamma",&lambda0*T,n+1)),1);
  if (&H1="two.sided" and n>=&lambda0*T) then
    pvalue=min(2*cdf("gamma",&lambda0*T,n),1);
run;

proc print data=test noobs;
run;

```

T	n	pvalue
1978	12	0.043064

```

/*Exact Poisson Confidence Interval*/

%let conf_level=95; *choices: 90, 95, 99, etc.;

data CI (drop=alpha);
set cases;
alpha=(1-0.01*&conf_level);
lambda_hat=n/T;
lambdaL=gaminv(alpha/2,n)/T;
lambdaU=gaminv(1-alpha/2,n+1)/T;

```

```
run;

proc print data=CI noobs;
run;
```

T	n	lambda_hat	lambdaL	lambdaU
1978	12	.006066734	.00313477	0.010597

In R:

```
#exact Poisson test
poisson.test(12, 1978, r=0.01, alternative="less")
```

Exact Poisson test

```
data: 12 time base: 1978
number of events = 12, time base = 1978, p-value = 0.04306
alternative hypothesis: true event rate is less than 0.01
95 percent confidence interval:
 0.0000000000 0.009829408
sample estimates:
 event rate
0.006066734
```

```
#exact Poisson confidence interval
install.packages("DescTools")
library(DescTools)
PoissonCI(12, 1978, method="exact", conf.level = 0.95)
```

```
      est      lwr.ci      upr.ci
0.006066734 0.00313477 0.01059736
```

Note that in the SAS code we used the following result to compute the p -value of the test as well as to compute the end-points of the confidence interval.

Lemma. The following identity is true:

$$\sum_{k=0}^{n-1} \frac{(\lambda T)^k}{k!} e^{-\lambda T} = \int_{\lambda T}^{\infty} \frac{x^{n-1}}{(n-1)!} e^{-x} dx.$$

The expression on the right is the upper-tail for a gamma distribution. Indeed, it is $\mathbb{P}(X > \lambda T)$ where X has a gamma distribution with parameters n and 1.

Proof of the Lemma: We use the integration by parts approach to compute iteratively

$$\begin{aligned} \int_{\lambda T}^{\infty} \frac{x^{n-1}}{(n-1)!} e^{-x} dx &= -\frac{x^{n-1}}{(n-1)!} e^{-x} \Big|_{\lambda T}^{\infty} + \int_{\lambda T}^{\infty} \frac{x^{n-2}}{(n-2)!} e^{-x} dx \\ &= \frac{(\lambda T)^{n-1}}{(n-1)!} e^{-\lambda T} + \int_{\lambda T}^{\infty} \frac{x^{n-2}}{(n-2)!} e^{-x} dx \\ &= \frac{(\lambda T)^{n-1}}{(n-1)!} e^{-\lambda T} + \frac{(\lambda T)^{n-2}}{(n-1)!} e^{-\lambda T} + \int_{\lambda T}^{\infty} \frac{x^{n-3}}{(n-3)!} e^{-x} dx \\ &= \dots = \sum_{k=0}^{n-1} \frac{(\lambda T)^k}{k!} e^{-\lambda T}. \quad \square \end{aligned}$$